3 Rekker

3.1 Innledning

> $restart:$

Maple skriver opp en tallfølge ved kommandoen seq.

- seq($a(n)$,n =1..5) skriver tallfølgen $a(1), a(2), a(3), a(4), a(5)$

> $seq(n, n = 1..20)$

$$1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20$$

Her lister Maple opp de 20 første heltallene.

En tallfølge der hvert av tallene i tallfølgen er summen av de to foregående tallene kalles Fibonacci-tallene. Disse tallene kan vi uttrykke ved følgende ligning.

> $a(n) = a(n-2) + a(n-1)$

$$a(n) = a(n-2) + a(n-1)$$

Det er for tungvint å finne mange tall i tallfølgen ved å summere for hånd. Ved hjelp av Maple går regnearbeidet hurtig.

Vi definerer først $a(n)$ som en funksjon av n, så gir vi $a(1)$ verdien 0 og $a(2)$ verdien 1. Deretter gjør Maple resten av jobben ved bruk av seq- kommandoen.

> $a := n \mapsto a(n-2) + a(n-1)$

$$a := n \mapsto a(n-2) + a(n-1)$$

> $a(1) := 0;\ a(2) := 1$

$$a(1) := 0$$

$$a(2) := 1$$

De 20 første Fibonacci-tallene blir da.

> $seq(a(n), n = 1..20)$

$$0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181$$

NB! For større verdier av n tar det lang tid å beregne tallene, fordi dette ikke er en effektiv måte å beregne fibonacci-tallene på.

> $a(30)$

$$514229$$

Maplekommandoen fibonacci(n) beregner det n'te Fibonacci-tallet

> $fibonacci(29), fibonacci(150)$

$$514229, 9969216677189303386214405760200$$

> $restart:$

Maple er også i stand til å løse ligningen

> $lign := a(n) = a(n-1) + a(n-2) : \%$

$$a(n) = a(n-1) + a(n-2)$$

eksakt.

> $rsolve(\{lign, a(1) = 0, a(2) = 1\}, a(n))$

$$\left(\frac{1}{2} - \frac{\sqrt{5}}{10}\right)\left(\frac{\sqrt{5}}{2} + \frac{1}{2}\right)^n + \left(\frac{1}{2} + \frac{\sqrt{5}}{10}\right)\left(-\frac{\sqrt{5}}{2} + \frac{1}{2}\right)^n$$

> $a := unapply(\%, n):$

Selv om denne eksakte løsningen inneholder rotuttrykk , så er alle fibonacci-tallene hele tall.

> $seq(simplify(a(n)), n = 1..50)$

0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765, 10946, 17711, 28657, 46368, 75025, 121393, 196418, 317811, 514229, 832040, 1346269, 2178309, 3524578, 5702887, 9227465, 14930352, 24157817, 39088169, 63245986, 102334155, 165580141, 267914296, 433494437, 701408733, 1134903170, 1836311903, 2971215073, 4807526976, 7778742049

Lar vi n bli større og større vil forholdet mellom hvert tall og det foregående nærme seg en bestemt verdi som kalles **det gyldne snitt**.

> $seq\left(evalf\left(\frac{a(n)}{a(n+1)}\right), n = 1..40\right)$

$5.000000007 \times 10^{-10}$, 0.9999999992, 0.5000000007, 0.6666666667, 0.6000000005, 0.6250000001, 0.6153846154, 0.6190476195, 0.6176470594, 0.6181818182, 0.6179775284, 0.6180555562, 0.6180257509, 0.6180371357, 0.6180327871, 0.6180344482, 0.6180338137, 0.6180340560, 0.6180339637, 0.6180339988, 0.6180339852, 0.6180339906, 0.6180339885, 0.6180339889, 0.6180339893, 0.6180339890, 0.6180339890, 0.6180339889, 0.6180339893, 0.6180339889, 0.6180339892, 0.6180339889, 0.6180339890, 0.6180339889, 0.6180339892, 0.6180339889, 0.6180339891, 0.6180339892, 0.6180339888, 0.6180339890

Tallet er

> $\frac{\sqrt{5}-1}{2} = evalf\left(\frac{\sqrt{5}-1}{2}\right)$

$$\frac{\sqrt{5}}{2} - \frac{1}{2} = 0.6180339880$$

3.2 Aritmetiske rekker

> $restart:$

Summerer vi leddene i en tallfølge får vi en rekke. Rekken kalles **aritmetisk** når differensen mellom hvert ledd og det foregående er konstant. Det allmenne leddet i en aritmetisk rekke er gitt ved

> $an := n \mapsto a_1 + (n-1) \cdot d:$

> $a[n] = an(n)$

$$a_n = a_1 + (n-1)\, d$$

Vi kan summere leddene i en rekke med add

- add$(a(n), n = n_1..n_2)$ summerer en sekvens av $a(n)$

Hvis det er mulig å finne en formel for en sum, må vi bruke kommandoen sum.

- sum$(a(n), n = n_1..n_2)$ gjør samme jobben som add, men er også i stand til å finne et symbolsk uttrykk (formel)

- AritmetiskRekke beregner det $n$te leddet a_n og summen s_n i en aritmetisk rekke

Eksempel 3.2.1

I en aritmetisk rekke er $a_5 = 5$ og $a_{17} = 8$

Finn a_1 og d .

Løsning

Vi definerer det n'te leddet ved

> $an := n \mapsto a[1] + (n-1) \cdot d :$

> $lign1 := an(5) = 5 : \%$

$$a_1 + 4\,d = 5$$

> $lign2 := an(17) = 8 : \%$

$$a_1 + 16\,d = 8$$

Vi løser ligningene.

> $solve(\{lign1, lign2\}, \{d, a_1\})$

$$\left\{d = \frac{1}{4}, a_1 = 4\right\}$$

> $AritmetiskRekke(a_1, d, n)$

$$a_n = a_1 + (n-1)\,d$$

$$s_n = \frac{\left(2\,a_1 + (n-1)\,d\right)\,n}{2}$$

> $a[5], a[17] := 5, 8 :$

> $AritmetiskRekke(a[1], d, 5)$

$$5 = a_1 + 4\,d$$

$$s_5 = 5\,a_1 + 10\,d$$

> $AritmetiskRekke(a[1], d, 17)$

$$8 = a_1 + 16\,d$$

$$s_{17} = 17\,a_1 + 136\,d$$

> $solve(\{5 = a_1 + 4\,d, 8 = a_1 + 16\,d\})$

$$\left\{d = \frac{1}{4}, a_1 = 4\right\}$$

Vi kan også finne summen av de n første leddene i en aritmetisk rekke ved

> $S[n] = \sum_{i=1}^{n} an(i)$

$$S_n = \frac{d\,(n+1)^2}{2} - \frac{3\,d\,(n+1)}{2} + a_1\,(n+1) - a_1 + d$$

> $lign := simplify(\%) : \%$

$$S_n = \frac{\left(2\,a_1 + (n-1)\,d\right)\,n}{2}$$

For å få den kjente formelen for S_n løser vi

> $a[n] = an(n)$

$$a_n = a_1 + (n-1)\,d$$

med hensyn på d og putter uttrykket inn i $S_n = a_1\, n + \frac{1}{2}\, d\, n^2 - \frac{1}{2}\, d\, n$.

> $solve(\%, \{d\})$

$$\left\{ d = -\frac{a_1 - a_n}{n - 1} \right\}$$

> $subs(\%, lign)$

$$S_n = \frac{(a_1 + a_n)\, n}{2}$$

>

Eksempel 3.2.2

a) Finn summen av de 75 første leddene i rekka $7 + 15 + 23 + \cdots$

b) Finn summen av de n første naturlige tallene.

Løsning

a) Her er $a_1 = 7$ og $d = 8$. Det gir

> $a[75] = 7 + (75 - 1) \cdot 8$

$$a_{75} = 599$$

Summen blir:

> $S[75] = \frac{(7 + 599) \cdot 75}{2}$

$$S_{75} = 22725$$

> $AritmetiskRekke(7, 8, 75)$

$$a_{75} = 599$$

$$s_{75} = 22725$$

Direkte beregning

> $S[75] = add(7 + (i - 1) \cdot 8, i = 1 ..75)$

$$S_{75} = 22725$$

b)

Her bruker vi Maples sum direkte for å finne formelen.

> $S[n] = \sum_{i=1}^{n} i$

$$S_n = \frac{(n + 1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

> $factor(\%)$

$$S_n = \frac{n\,(n + 1)}{2}$$

Hvis vi bruker Sum med stor S, får vi skrevet det symbolske uttrykket for en sum.

> $Sum(i, i = 1 ..n) = sum(i, i = 1 ..n)$

$$\sum_{i=1}^{n} i = \frac{(n + 1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

> $factor(\%)$

$$\sum_{i=1}^{n} i = \frac{n\,(n+1)}{2}$$

Eksempel 3.2.3

a) Finn summen av de seksti første partallene.

b) Finn summen av de hundre første oddetallene.

Løsning

a)

Her skal vi la Maple gjøre jobben direkte og uten bruk av formel for summen av en aritmetisk rekke.

> $Sum(2\,i, i = 1\,..60) = sum(2\,i, i = 1\,..60)$

$$\sum_{i=1}^{60} 2\,i = 3660$$

og generelt

> $Sum(2\,i, i = 1\,..n) = sum(2\,i, i = 1\,..n)$

$$\sum_{i=1}^{n} 2\,i = (n+1)^2 - n - 1$$

> $factor(\%)$

$$\sum_{i=1}^{n} 2\,i = n\,(n+1)$$

>

b)

Summen av oddetallene blir

> $Sum(2\,i - 1, i = 1\,..100) = sum(2\,i - 1, i = 1\,..100)$

$$\sum_{i=1}^{100} (2\,i - 1) = 10000$$

og generelt

> $Sum(2\,i - 1, i = 1\,..n) = sum(2\,i - 1, i = 1\,..n)$

$$\sum_{i=1}^{n} (2\,i - 1) = (n+1)^2 - 2\,n - 1$$

> $factor(\%)$

$$\sum_{i=1}^{n} (2\,i - 1) = n^2$$

3.3 Geometriske rekker

Vi har en geometrisk rekke når forholdet mellom hvert ledd og det foregående er konstant.

Det allmenne leddet er gitt ved $a_n = a_1 k^{n-1}$

Bruker vi dette allmenne leddet og sum-kommandoen, finner Maple summen for oss. I formelen under lar vi summasjonsindeksen være i.

> $Sum\left(a[1] \cdot k^{i-1}, i = 1\,..n\right) = sum\left(a[1] \cdot k^{i-1}, i = 1\,..n\right)$

$$\sum_{i=1}^{n} a_1 k^{i-1} = \left(\frac{k^{n+1}}{k(k-1)} - \frac{1}{k-1} \right) a_1$$

> *simplify*(%)

$$a_1 \left(\sum_{i=1}^{n} k^{i-1} \right) = \frac{(-1+k^n)\, a_1}{k-1}$$

Med GeometriskRekke får vi

> *GeometriskRekke*($a[1], k, n$)

$$a_n = a_1 k^{n-1}$$

$$s_n = \frac{(-1+k^n)\, a_1}{k-1}$$

Eksempel 3.3.1

Hva er det sjuende leddet i en geometrisk rekke der de to første leddene er 6 og 2.

Løsning

Det n'te leddet er gitt ved

> *restart* :

> $an := n \mapsto a[1] \cdot k^{n-1}$:

$a_n = 6\, an(n)$

$$a_n = 6\, a_1 k^{n-1}$$

> $a_1 := 6$

$$a_1 := 6$$

$a_2 = 2$ gir

> $an(2) = 2$

$$6k = 2$$

> $k := solve(\%, k)$

$$k := \frac{1}{3}$$

> $a[7] = an(7)$

$$a_7 = \frac{2}{243}$$

> *GeometriskRekke*$\left(6, \frac{1}{3}, 7\right)$

$$a_7 = \frac{2}{243}$$

$$s_7 = \frac{2186}{243}$$

>

Eksempel 3.3.2

Regn ut summen av de 13 første leddene i rekken $8 + 16 + 32 + 64 + 128 + \cdots$

Løsning

Her er $a_1 = 8$, $k = 2$ og antall ledd $n = 13$. Det gir

> *restart* :

> *GeometriskRekke*(8, 2, 13)

$$a_{13} = 32768$$

$$s_{13} = 65528$$

Ved sum-kommandoen

> $Sum(2^n, n = 3..15) = sum(2^n, n = 3..15)$

$$\sum_{n=3}^{15} 2^n = 65528$$

Eksempel 3.3.3

Hvor mye vokser 15000 kroner til etter sju år når renten er 7,5 % per år.

Løsning

Vi definerer kapitalen K etter n år som en funksjon av n ved formelen

> $Kn := n \mapsto K[0] \cdot \left(1 + \frac{p}{100}\right)^n$:

$K[n] = Kn(n)$

$$K_n = K_0 \left(1 + \frac{p}{100}\right)^n$$

> $K[0], p := 15000, 7.5$

$$K_0, p := 15000, 7.5$$

> $K[7] = Kn(7)$

$$K_7 = 24885.73710$$

Når vi har definert K som en funksjon av n , kan vi bruke seq-kommandoen for å få beregnet en hel serie med K-verdier, for eksempel alle verdiene i løpet av de første 20 årene..

> $seq(K[n] = Kn(n), n = 1..20)$

$K_1 = 16125.00000$, $K_2 = 17334.37500$, $K_3 = 18634.45312$, $K_4 = 20032.03712$, $K_5 = 21534.43989$, $K_6 = 23149.52289$, $K_7 = 24885.73710$, $K_8 = 26752.16739$, $K_9 = 28758.57993$, $K_{10} = 30915.47343$, $K_{11} = 33234.13394$, $K_{12} = 35726.69398$, $K_{13} = 38406.19604$, $K_{14} = 41286.66074$, $K_{15} = 44383.16030$, $K_{16} = 47711.89731$, $K_{17} = 51290.28962$, $K_{18} = 55137.06134$, $K_{19} = 59272.34094$, $K_{20} = 63717.76650$

>

3.4 Konvergens og uendelige rekker

En uendelig geometrisk rekke $a_1 + a_2 + \cdots + a_n + \cdots$

konvergerer når $k = \left|\frac{a_n}{a_{n-1}}\right| < 1$ for $n = 2, 3, \cdots$

Vi kan la Maple finne summen for en konvergerende, uendelig geometrisk rekke. Vi finner først summen av en endelig geometrisk rekke.

> *restart* :

> $S := n \to Sum(a[1] \cdot k^i, i = 0 .. n)$:

> $eq := S(n) = value(S(n))$: %

$$\sum_{i=0}^{n} a_1 k^i = \left(\frac{k^{n+1}}{k-1} - \frac{1}{k-1} \right) a_1$$

> *factor*(%)

$$\sum_{i=0}^{n} a_1 k^i = \frac{(k^{n+1} - 1)\, a_1}{k-1}$$

Vi må først fortelle Maple at $-1 < k < 1$. Det gjør vi ved assume.

- assume(*betingelse*) antar at *betingelse* er oppfylt

> $assume(k < 1, k > -1)$:

> $S(\infty) = value(S(\infty))$

$$\sum_{i=0}^{\infty} a_1 k^i = -\frac{a_1}{k-1}$$

Eksempel 3.4.1

Vis at den uendelige geometriske rekken $\frac{1}{3} - \frac{1}{9} + \frac{1}{27} - \frac{1}{81} + \cdots$

konvergerer og finn summen.

Løsning

Alle nevnerne kan skrives som potenser av 3. Det kan vi vise ved ifactor

Vi skriver tallene på listeform (i en hakeparentes)

> $L := \left[\frac{1}{3}, -\frac{1}{9}, \frac{1}{27}, -\frac{1}{81}, \frac{1}{243} \right]$:

> $ifactor(L)$

$$\left[\frac{1}{(3)}, -\frac{1}{(3)^2}, \frac{1}{(3)^3}, -\frac{1}{(3)^4}, \frac{1}{(3)^5} \right]$$

Nå ser vi lett at forholdet mellom hvert ledd og det foregående er $-\frac{1}{3}$. Det betyr at rekken konvergerer. Summen kan finnes av

> $S = \frac{1}{3\left(1 - \left(-\frac{1}{3}\right)\right)}$

$$S = \frac{1}{4}$$

Vi lar så Maple finne summen direkte ved å summere uendelig mange ledd. Det allmenne leddet i rekken kan skrives

> $Sum\left(\frac{(-1)^{n+1}}{3^n}, n = 1 .. \infty\right) = sum\left(\frac{(-1)^{n+1}}{3^n}, n = 1 .. \infty\right)$

$$\sum_{n=1}^{\infty} \frac{(-1)^{n+1}}{3^n} = \frac{1}{4}$$

>